

## Geodesiv BVP

Consider a geodesic that connects two points  $P_i$  and  $P_j$  with coordinates  $x_i^a$  and  $x_j^a$ . Our aim is to construct a solution  $x^a(s)$  of the geodesic equation such that  $x^a(0) = x_i^a$  and  $x^a(1) = x_j^a$ .

We will do this in two stages. First we will solve

$$x_j^a = x_i^a + y^a - \sum_{k=2}^{\infty} \frac{1}{k!} \Gamma_{\underline{b}_k}^a y^{\underline{b}_k}$$

for  $y^a$  as an explicit polynomial in  $x_i^a$  and  $x_j^a$ . The functions  $\Gamma_{\underline{b}_k}^a$  are the generalised connections for the  $x^a$  RNC frame evaluated at  $x^a = x_i^a$ .

In the second stage, we will substitute our expression for  $y^a$  into

$$x^a(s) = x_i^a + sy^a - \sum_{k=2}^{\infty} \frac{1}{k!} \Gamma_{\underline{b}_k}^a y^{\underline{b}_k} s^k$$

to obtain the desired solution to the two point boundary value problem.

## Algorithm

First we rewrite the main equation in the suggestive form

$$y^a = x_j^a - x_i^a + \sum_{k=2}^{\infty} \frac{1}{k!} \Gamma_{\underline{b}_k}^a y^{\underline{b}_k}$$

and then use this as the basis of a fixed point iteration scheme. Start with the first approximation  $y_1^a = Dx^a$ , where  $Dx^a = x_j^a - x_i^a$ , then compute the successive approximations

$$y_2^a = Dx^a + \frac{1}{2!} \Gamma_{bc}^a y_1^b y_1^c$$

$$y_3^a = Dx^a + \frac{1}{2!} \Gamma_{bc}^a y_2^b y_2^c + \frac{1}{3!} \Gamma_{bcd}^a y_1^b y_1^c y_1^d$$

$$y_4^a = Dx^a + \frac{1}{2!} \Gamma_{bc}^a y_3^b y_3^c + \frac{1}{3!} \Gamma_{bcd}^a y_2^b y_2^c y_2^d + \frac{1}{4!} \Gamma_{bcde}^a y_1^b y_1^c y_1^d y_1^e$$

$$y_5^a = Dx^a + \frac{1}{2!} \Gamma_{bc}^a y_4^b y_4^c + \frac{1}{3!} \Gamma_{bcd}^a y_3^b y_3^c y_3^d + \frac{1}{4!} \Gamma_{bcde}^a y_2^b y_2^c y_2^d y_2^e + \frac{1}{5!} \Gamma_{bcdef}^a y_1^b y_1^c y_1^d y_1^e y_1^f$$

and so on.

Our aim in these notes is to produce approximations (of the metric, the connection, geodesic length etc.) that are accurate to a specified order in the curvature. We assign orders to the curvatures according to the following scheme. Let  $\epsilon$  be the order parameter, then we set

$$R^a{}_{bcd} = \mathcal{O}(\epsilon^2), \quad \nabla_e R^a{}_{bcd} = \mathcal{O}(\epsilon^3)$$

$$\nabla_{ef} R^a{}_{bcd} = \mathcal{O}(\epsilon^4), \quad \nabla_{efg} R^a{}_{bcd} = \mathcal{O}(\epsilon^5), \quad \nabla_{efgh} R^a{}_{bcd} = \mathcal{O}(\epsilon^6) \quad \dots$$

Note that as the  $\Gamma^a_{b_k}$  are polynomial functions of the curvatures they will inherit a similar dependence on the order parameter  $\epsilon$ . By inspection we find  $\Gamma^a_{b_k} = \mathcal{O}(\epsilon^k)$ .

We will retain terms up to and including  $\mathcal{O}(\epsilon^5)$ . At various points in the following calculations we will truncate intermediate results to  $\mathcal{O}(\epsilon^5)$ . My experience is that this is crucial to getting reasonable computational times. I found that by not imposing this truncation condition the results would blow out beyond my patience (I want calculations of the order of minutes and in ten's of Mbytes of memory, yes, I ask for a lot!).

```
::KeepHistory(false).
::PostDefaultRules( @@collect_terms!(%), @@sumflatten!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,t,u#,v#}::Indices.

# --- the Q are shorthand for the genGamma, saves typing -----
{Q^{a}_{b c},Q^{a}_{b c d},Q^{a}_{b c d e},Q^{a}_{b c d e f},x^{a},Dx^{a}}::SortOrder.

Q^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).
Q^{a}_{b c d}::TableauSymmetry(shape={3}, indices={1,2,3}).
Q^{a}_{b c d e}::TableauSymmetry(shape={4}, indices={1,2,3,4}).
Q^{a}_{b c d e f}::TableauSymmetry(shape={5}, indices={1,2,3,4,5}).

\nabla_{\#}::PartialDerivative.

R_{a b c d}::RiemannTensor.
R^{a}_{b c d}::RiemannTensor.

# =====
#   the first 5 approximations to y^a
# =====

# --- y01 -----
y01:= Dx^{a};
```

```

# --- y02 -----
y02:= Dx^{a} + 1/2 Q^{a}_{b c} y01^{b} y01^{c}:

@substitute!(y02)( y01^{a} -> @(y01) ): @distribute!(%);

# --- y03 -----
y03:= Dx^{a} + 1/2 Q^{a}_{b c} y02^{b} y02^{c}
      + 1/6 Q^{a}_{b c d} y01^{b} y01^{c} y01^{d}:

@substitute!(y03)( y01^{a} -> @(y01),
                  y02^{a} -> @(y02) ): @distribute!(%):

# --- truncate to 3rd order in Dx^a

poly:=@(y03):

Dx^{a}::Weight(label=Dxterms,value=1).

term00:=@(poly): @keep_weight!(term00){Dxterms}{0}:
term01:=@(poly): @keep_weight!(term01){Dxterms}{1}:
term02:=@(poly): @keep_weight!(term02){Dxterms}{2}:
term03:=@(poly): @keep_weight!(term03){Dxterms}{3}:

y03:=@(term00) + @(term01) + @(term02) + @(term03):
@prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

# --- y04 -----
y04:= Dx^{a} + 1/2 Q^{a}_{b c} y03^{b} y03^{c}
      + 1/6 Q^{a}_{b c d} y02^{b} y02^{c} y02^{d}
      + 1/24 Q^{a}_{b c d e} y01^{b} y01^{c} y01^{d} y01^{e}:

@substitute!(y04)( y03^{a} -> @(y03),
                  y02^{a} -> @(y02),
                  y01^{a} -> @(y01) ): @distribute!(%):

# --- truncate to 4th order in Dx^a

poly:=@(y04):

```

```

Dx^{a}::Weight(label=Dxterms,value=1).

term00:=@(poly): @keep_weight!(term00){Dxterms}{0}:
term01:=@(poly): @keep_weight!(term01){Dxterms}{1}:
term02:=@(poly): @keep_weight!(term02){Dxterms}{2}:
term03:=@(poly): @keep_weight!(term03){Dxterms}{3}:
term04:=@(poly): @keep_weight!(term04){Dxterms}{4}:

y04:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04):
@prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

# --- y05 -----
y05:= Dx^{a} + 1/2 Q^{a}_{b c} y04^{b} y04^{c}
      + 1/6 Q^{a}_{b c d} y03^{b} y03^{c} y03^{d}
      + 1/24 Q^{a}_{b c d e} y02^{b} y02^{c} y02^{d} y02^{e}
      + 1/120 Q^{a}_{b c d e f} y01^{b} y01^{c} y01^{d} y01^{e} y01^{f}:

@substitute!(y05)( y04^{a} -> @(y04),
                  y03^{a} -> @(y03),
                  y02^{a} -> @(y02),
                  y01^{a} -> @(y01) ): @distribute!(%):

# --- truncate to 5th order in Dx^a

poly:=@(y05):

Dx^{a}::Weight(label=Dxterms,value=1).

term00:=@(poly): @keep_weight!(term00){Dxterms}{0}:
term01:=@(poly): @keep_weight!(term01){Dxterms}{1}:
term02:=@(poly): @keep_weight!(term02){Dxterms}{2}:
term03:=@(poly): @keep_weight!(term03){Dxterms}{3}:
term04:=@(poly): @keep_weight!(term04){Dxterms}{4}:
term05:=@(poly): @keep_weight!(term05){Dxterms}{5}:

y05:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04) + @(term05):
@prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

# =====

```

```
# the first 5 approximations to y^a
# =====
```

$$y01 := Dx^a$$

$$y02 := Dx^a + \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c$$

$$y03 := Dx^a + \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Dx^c Dx^d Dx^e + \frac{1}{6} Q^a{}_{bcd} Dx^b Dx^c Dx^d$$

$$y04 := Dx^a + \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Dx^c Dx^d Dx^e + \frac{1}{6} Q^a{}_{bcd} Dx^b Dx^c Dx^d + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Q^d{}_{fg} Dx^c Dx^e Dx^f Dx^g \\ + \frac{1}{6} Q^a{}_{bc} Q^b{}_{def} Dx^c Dx^d Dx^e Dx^f + \frac{1}{8} Q^a{}_{bc} Q^b{}_{de} Q^c{}_{fg} Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} Q^b{}_{cd} Q^a{}_{bef} Dx^c Dx^d Dx^e Dx^f + \frac{1}{24} Q^a{}_{bcde} Dx^b Dx^c Dx^d Dx^e$$

$$y05 := Dx^a + \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Dx^c Dx^d Dx^e + \frac{1}{6} Q^a{}_{bcd} Dx^b Dx^c Dx^d + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Q^d{}_{fg} Dx^c Dx^e Dx^f Dx^g \\ + \frac{1}{6} Q^a{}_{bc} Q^b{}_{def} Dx^c Dx^d Dx^e Dx^f + \frac{1}{8} Q^a{}_{bc} Q^b{}_{de} Q^c{}_{fg} Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} Q^b{}_{cd} Q^a{}_{bef} Dx^c Dx^d Dx^e Dx^f + \frac{1}{24} Q^a{}_{bcde} Dx^b Dx^c Dx^d Dx^e \\ + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Q^d{}_{fg} Q^f{}_{hi} Dx^c Dx^e Dx^g Dx^h Dx^i + \frac{1}{6} Q^a{}_{bc} Q^b{}_{de} Q^d{}_{fgh} Dx^c Dx^e Dx^f Dx^g Dx^h + \frac{1}{8} Q^a{}_{bc} Q^b{}_{de} Q^d{}_{fg} Q^e{}_{hi} Dx^c Dx^f Dx^g Dx^h Dx^i \\ + \frac{1}{4} Q^a{}_{bc} Q^d{}_{ef} Q^b{}_{dgh} Dx^c Dx^e Dx^f Dx^g Dx^h + \frac{1}{24} Q^a{}_{bc} Q^b{}_{defg} Dx^c Dx^d Dx^e Dx^f Dx^g + \frac{1}{4} Q^a{}_{bc} Q^b{}_{de} Q^c{}_{fg} Q^d{}_{hi} Dx^e Dx^f Dx^g Dx^h Dx^i \\ + \frac{1}{12} Q^a{}_{bc} Q^b{}_{de} Q^c{}_{fgh} Dx^d Dx^e Dx^f Dx^g Dx^h + \frac{1}{4} Q^b{}_{cd} Q^c{}_{ef} Q^a{}_{bgh} Dx^d Dx^e Dx^f Dx^g Dx^h + \frac{1}{12} Q^a{}_{bcd} Q^b{}_{efg} Dx^c Dx^d Dx^e Dx^f Dx^g \\ + \frac{1}{8} Q^b{}_{cd} Q^e{}_{fg} Q^a{}_{beh} Dx^c Dx^d Dx^f Dx^g Dx^h + \frac{1}{12} Q^b{}_{cd} Q^a{}_{befg} Dx^c Dx^d Dx^e Dx^f Dx^g + \frac{1}{120} Q^a{}_{bcdef} Dx^b Dx^c Dx^d Dx^e Dx^f$$

```
# =====
# now truncate y05 to 5-th order in the curvatures
# =====
# Q^{a}_{...} with n-lower indices contributes \eps^n in the curvature expansion.
# We will retain terms to \eps^5
```

```

poly:=@(y05):

Q^{a}_{b c}::Weight(label=Qterms,value=2).
Q^{a}_{b c d}::Weight(label=Qterms,value=3).
Q^{a}_{b c d e}::Weight(label=Qterms,value=4).
Q^{a}_{b c d e f}::Weight(label=Qterms,value=5).

term00:=@(poly): @keep_weight!(term00){Qterms}{0}:
term01:=@(poly): @keep_weight!(term01){Qterms}{1}:
term02:=@(poly): @keep_weight!(term02){Qterms}{2}:
term03:=@(poly): @keep_weight!(term03){Qterms}{3}:
term04:=@(poly): @keep_weight!(term04){Qterms}{4}:
term05:=@(poly): @keep_weight!(term05){Qterms}{5}:

y05:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04) + @(term05):
@prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

# =====
#   the final approximation y05 for y^a expressed in terms of the Gammas
# =====


$$\begin{aligned}
y05 := & Dx^a + \frac{1}{2} Q^a_{bc} Dx^b Dx^c + \frac{1}{6} Q^a_{bcd} Dx^b Dx^c Dx^d + \frac{1}{2} Q^a_{bc} Q^b_{de} Dx^c Dx^d Dx^e + \frac{1}{24} Q^a_{bcde} Dx^b Dx^c Dx^d Dx^e \\
& + \frac{1}{6} Q^a_{bc} Q^b_{def} Dx^c Dx^d Dx^e Dx^f + \frac{1}{4} Q^b_{cd} Q^a_{bef} Dx^c Dx^d Dx^e Dx^f + \frac{1}{120} Q^a_{bcdef} Dx^b Dx^c Dx^d Dx^e Dx^f
\end{aligned}$$


# --- imported from geodesic-ivp.lib -----

genGamma02=="import geodesic-ivp.lib genGamma02":
@run(genGamma02){"/Users/leo/local/sh/cdbfile"}:

genGamma03=="import geodesic-ivp.lib genGamma03":
@run(genGamma03){"/Users/leo/local/sh/cdbfile"}:

genGamma04=="import geodesic-ivp.lib genGamma04":
@run(genGamma04){"/Users/leo/local/sh/cdbfile"}:

genGamma05=="import geodesic-ivp.lib genGamma05":
@run(genGamma05){"/Users/leo/local/sh/cdbfile"}:

```

```

genGamma06:="import geodesic-ivp.lib genGamma06":
@run(genGamma06){"/Users/leo/local/sh/cdbfile"}:

# by inspection we see that y05 contains terms linear in  $Q^{\{a\}}_{\{...\}}$  and three quadratic terms
#  $Q^{\{a\}}_{\{bc\}}Q^{\{b\}}_{\{de\}}$ ,  $Q^{\{a\}}_{\{bc\}}Q^{\{b\}}_{\{def\}}$  and  $Q^{\{b\}}_{\{cd\}}Q^{\{a\}}_{\{bef\}}$ 
# these quadratic terms will introduce unwanted higher order curvature terms

# save this version of y05 as it will be used later when we compute  $x^a(s)$ , the  $x^a$  of the geodesic

y01Gamma:=@(y01); "y01Gamma.txt"
y02Gamma:=@(y02); "y02Gamma.txt"
y03Gamma:=@(y03); "y03Gamma.txt"
y04Gamma:=@(y04); "y04Gamma.txt"
y05Gamma:=@(y05); "y05Gamma.txt"

@substitute!(y05)( $Q^{\{a\}}_{\{b\}c}Q^{\{b\}}_{\{d\}e}$  -> QQ1 $^{\{a\}}_{\{c\}d}e$ ,
                  $Q^{\{a\}}_{\{b\}c}Q^{\{b\}}_{\{d\}e}f$  -> QQ2 $^{\{a\}}_{\{c\}d}ef$ ,
                  $Q^{\{b\}}_{\{c\}d}Q^{\{a\}}_{\{b\}ef}$  -> QQ3 $^{\{a\}}_{\{c\}d}ef$ ):

QQ1:= $Q^{\{a\}}_{\{b\}c}Q^{\{b\}}_{\{d\}e}$ :
QQ2:= $Q^{\{a\}}_{\{b\}c}Q^{\{b\}}_{\{d\}e}f$ :
QQ3:= $Q^{\{b\}}_{\{c\}d}Q^{\{a\}}_{\{b\}ef}$ :

# clear the factor  $B^{\{a\}}$  in the genGamma0*

@substitute!(genGamma02)( $B^{\{a\}}$  -> 1):
@substitute!(genGamma03)( $B^{\{a\}}$  -> 1):
@substitute!(genGamma04)( $B^{\{a\}}$  -> 1):
@substitute!(genGamma05)( $B^{\{a\}}$  -> 1):

# the genGamma0* imported from geodesic-ivp.cdbp are not symmetrized, do so now
# this is only required for QQ1, QQ2 and QQ3 as all other terms in y05 are symmetrized
# despite this I find the final result for y05 does not change regardless of whether or not
# the following @sym(...) are applied

@sym(genGamma02){_b}, _{c}}:
@sym(genGamma03){_b}, _{c}, _{d}}:

@substitute!(QQ1)( $Q^{\{a\}}_{\{b\}c}$  -> @(genGamma02)):
@substitute!(QQ2)( $Q^{\{a\}}_{\{b\}c}$  -> @(genGamma02),  $Q^{\{a\}}_{\{b\}cd}$  -> @(genGamma03)):
@substitute!(QQ3)( $Q^{\{a\}}_{\{b\}c}$  -> @(genGamma02),  $Q^{\{a\}}_{\{b\}cd}$  -> @(genGamma03)):

```

```

# --- to retain 5-th order curvature terms, we can truncate QQ1 at  $O(x^3)$  and QQ2,QQ3 at  $O(x^2)$  ---
# --- truncate QQ1 to include  $O(x^3)$ 
poly:=@(QQ1):
@distributed!(%):

x^{a}::Weight(label=xterms,value=1).

term00:=@(poly): @keep_weight!(term00){xterms}{0}:
term01:=@(poly): @keep_weight!(term01){xterms}{1}:
term02:=@(poly): @keep_weight!(term02){xterms}{2}:
term03:=@(poly): @keep_weight!(term03){xterms}{3}:

QQ1:=@(term00) + @(term01) + @(term02) + @(term03); "QQ1.txt"
# --- truncate QQ2 to include  $O(x^2)$ 
poly:=@(QQ2):
@distributed!(%):

x^{a}::Weight(label=xterms,value=1).

term00:=@(poly): @keep_weight!(term00){xterms}{0}:
term01:=@(poly): @keep_weight!(term01){xterms}{1}:
term02:=@(poly): @keep_weight!(term02){xterms}{2}:

QQ2:=@(term00) + @(term01) + @(term02); "QQ2.txt"
# --- truncate QQ3 to include  $O(x^2)$ 
poly:=@(QQ3):
@distributed!(%):

x^{a}::Weight(label=xterms,value=1).

term00:=@(poly): @keep_weight!(term00){xterms}{0}:
term01:=@(poly): @keep_weight!(term01){xterms}{1}:
term02:=@(poly): @keep_weight!(term02){xterms}{2}:

QQ3:=@(term00) + @(term01) + @(term02); "QQ3.txt"

```



```

@substitute!(y05)(Q^{a}_{b c} -> @(genGamma02),
                  Q^{a}_{b c d} -> @(genGamma03),
                  Q^{a}_{b c d e} -> @(genGamma04),
                  Q^{a}_{b c d e f} -> @(genGamma05),
                  QQ1^{a}_{c d e} -> @(QQ1),
                  QQ2^{a}_{c d e f} -> @(QQ2),
                  QQ3^{a}_{c d e f} -> @(QQ3)):

@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

y05Riemann:=@(y05); "y05Riemann.txt"

# =====
#   the final approximation y05 for y^a expressed in terms of the Riemann curvatures
# =====

```

$$\begin{aligned}
y_{05} := & Dx^a + \frac{1}{3} R^a{}_{bcd} x^c Dx^b Dx^d + \frac{1}{12} \nabla_b R^a{}_{cde} x^c x^d Dx^b Dx^e + \frac{1}{6} \nabla_b R^a{}_{cde} x^b x^d Dx^c Dx^e + \frac{1}{24} \nabla^a R_{bcde} x^b x^d Dx^c Dx^e + \frac{4}{45} R^a{}_{bcd} R_{cefg} x^b x^d x^f Dx^e Dx^g \\
& - \frac{2}{45} R^a{}_{bcd} R_{cefg} x^d x^e x^f Dx^b Dx^g - \frac{1}{45} R^a{}_{bcd} R_{cefg} x^b x^e x^f Dx^d Dx^g + \frac{1}{20} \nabla_{bc} R^a{}_{def} x^b x^d x^e Dx^c Dx^f + \frac{1}{20} \nabla_{bc} R^a{}_{def} x^b x^c x^e Dx^d Dx^f \\
& + \frac{1}{45} R^a{}_{bcd} R_{befg} x^c x^e x^f Dx^d Dx^g + \frac{1}{40} \nabla^a{}_b R_{cdef} x^b x^c x^e Dx^d Dx^f + \frac{2}{45} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^f x^g x^h Dx^c Dx^e + \frac{1}{60} R^a{}_{bcd} \nabla_e R_{cfgh} x^b x^d x^f x^g Dx^e Dx^h \\
& + \frac{2}{45} R^a{}_{bcd} \nabla_e R_{cfgh} x^b x^d x^e x^g Dx^f Dx^h - \frac{1}{45} R^a{}_{bcd} \nabla_e R_{cfgh} x^d x^e x^f x^g Dx^b Dx^h - \frac{1}{90} R^a{}_{bcd} \nabla_e R_{cfgh} x^b x^e x^f x^g Dx^d Dx^h - \frac{1}{90} R_{bcde} \nabla_f R^a{}_{gbh} x^c x^d x^g x^h Dx^e Dx^f \\
& - \frac{1}{45} R_{bcde} \nabla_f R^a{}_{gbh} x^c x^d x^f x^h Dx^e Dx^g - \frac{1}{90} R_{bcde} \nabla_f R^a{}_{gbh} x^c x^d x^f x^g Dx^e Dx^h + \frac{1}{60} \nabla_{bcd} R^a{}_{efg} x^b x^c x^e x^f Dx^d Dx^g + \frac{1}{90} \nabla_{bcd} R^a{}_{efg} x^b x^c x^d x^f Dx^e Dx^g \\
& + \frac{1}{72} R^a{}_{bcd} \nabla_c R_{efgh} x^b x^d x^e x^g Dx^f Dx^h + \frac{1}{90} R^a{}_{bcd} \nabla_e R_{bfgh} x^c x^e x^f x^g Dx^d Dx^h - \frac{1}{90} R_{bcde} \nabla^a R_{bfgh} x^c x^d x^f x^g Dx^e Dx^h + \frac{1}{90} R_{bcde} \nabla_f R^a{}_{bgh} x^c x^d x^f x^g Dx^e Dx^h \\
& + \frac{1}{120} \nabla^a{}_b R_{defg} x^b x^c x^d x^f Dx^e Dx^g + \frac{1}{12} \nabla_b R^a{}_{cde} x^d Dx^b Dx^c Dx^e + \frac{1}{45} R^a{}_{bcd} R_{cefg} x^b x^f Dx^d Dx^e Dx^g + \frac{4}{45} R^a{}_{bcd} R_{cefg} x^d x^f Dx^b Dx^e Dx^g \\
& - \frac{1}{45} R^a{}_{bcd} R_{cefg} x^e x^f Dx^b Dx^d Dx^g + \frac{1}{20} \nabla_{bc} R^a{}_{def} x^b x^e Dx^c Dx^d Dx^f + \frac{1}{60} \nabla_{bc} R^a{}_{def} x^d x^e Dx^b Dx^c Dx^f + \frac{7}{45} R^a{}_{bcd} R_{befg} x^c x^f Dx^d Dx^e Dx^g \\
& + \frac{1}{120} \nabla^a{}_b R_{cdef} x^c x^e Dx^b Dx^d Dx^f + \frac{1}{54} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^g x^h Dx^c Dx^e Dx^f + \frac{1}{27} R_{bcde} \nabla_b R^a{}_{fgh} x^d x^f x^g Dx^c Dx^e Dx^h + \frac{1}{27} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^f x^h Dx^c Dx^e Dx^g \\
& + \frac{2}{27} R_{bcde} \nabla_f R^a{}_{bgh} x^d x^f x^g Dx^c Dx^e Dx^h - \frac{1}{108} R_{bcde} \nabla^a R_{bfgh} x^d x^f x^g Dx^c Dx^e Dx^h + \frac{1}{54} R^a{}_{bcd} \nabla_e R_{cfgh} x^d x^f x^g Dx^b Dx^e Dx^h \\
& + \frac{1}{27} R^a{}_{bcd} \nabla_e R_{cfgh} x^d x^e x^g Dx^b Dx^f Dx^h + \frac{1}{108} R^a{}_{bcd} \nabla_c R_{efgh} x^d x^e x^g Dx^b Dx^f Dx^h + \frac{1}{27} R^a{}_{bcd} \nabla_e R_{bfgh} x^c x^f x^g Dx^d Dx^e Dx^h \\
& + \frac{2}{27} R^a{}_{bcd} \nabla_e R_{bfgh} x^c x^e x^g Dx^d Dx^f Dx^h + \frac{1}{54} R^a{}_{bcd} \nabla_b R_{efgh} x^c x^e x^g Dx^d Dx^f Dx^h + \frac{1}{54} R^a{}_{bcd} R_{cefg} x^f Dx^b Dx^d Dx^e Dx^g + \frac{1}{216} R^a{}_{bcd} \nabla_e R_{cfgh} x^f x^g Dx^b Dx^d Dx^e Dx^h \\
& + \frac{1}{108} R^a{}_{bcd} \nabla_e R_{cfgh} x^e x^g Dx^b Dx^d Dx^f Dx^h + \frac{1}{432} R^a{}_{bcd} \nabla_c R_{efgh} x^e x^g Dx^b Dx^d Dx^f Dx^h + \frac{1}{216} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^g Dx^c Dx^e Dx^f Dx^h \\
& - \frac{5}{216} R_{bcde} \nabla_f R^a{}_{bgh} x^d x^g Dx^c Dx^e Dx^f Dx^h + \frac{1}{108} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^f Dx^c Dx^e Dx^g Dx^h + \frac{7}{108} R_{bcde} \nabla_b R^a{}_{fgh} x^d x^g Dx^c Dx^e Dx^f Dx^h \\
& + \frac{1}{216} R_{bcde} \nabla^a R_{bfgh} x^d x^g Dx^c Dx^e Dx^f Dx^h + \frac{1}{36} R_{bcde} \nabla_f R^a{}_{gbh} x^d x^h Dx^c Dx^e Dx^f Dx^g + \frac{1}{72} R^a{}_{bcd} \nabla_e R_{cfgh} x^d x^g Dx^b Dx^e Dx^f Dx^h \\
& + \frac{1}{24} R^a{}_{bcd} \nabla_e R_{bfgh} x^c x^g Dx^d Dx^e Dx^f Dx^h + \frac{1}{180} R^a{}_{bcd} \nabla_e R_{cfgh} x^g Dx^b Dx^d Dx^e Dx^f Dx^h + \frac{1}{120} R_{bcde} \nabla_f R^a{}_{gbh} x^d Dx^c Dx^e Dx^f Dx^g Dx^h
\end{aligned}$$

# =====  
# at this point we have a 5-th order approximation (in the curvature) for  $y^a$ , the direction  
# vector that ensures the geodesic passes through the points  $x^a$  (at  $s=0$ ) and  $x^a+Dx^a$  (at  $s=1$ ).  
#

```

# now we want an equation for  $x^a(s)$  along that geodesic for  $0 \leq s \leq 1$ 
# =====

y:=s @(y05Gamma):
@distribute!(y):

# --- approximations to  $y^a$  that ensure  $\backslash\text{BigO}\{\backslash\text{eps}^5\}$  accuracy in coord -----
# --- these were crafted by hand, after looking at printouts. maybe better to automate this

y01:=s Dx{a}:
y02:=@(y01) + s (1/2) Q{a}_{b c} Dx{b} Dx{c}:
y03:=@(y02) + s (1/6) Q{a}_{b c d} Dx{b} Dx{c} Dx{d}:
y05:=@(y):

# --- now compute the  $x^a$  as a function of  $\backslash\eta$ , this will solve the geodesic BVP -----

coord:=x{a} + y05{a} - (1/2) Q{a}_{b c} y03{b} y03{c}
- (1/6) Q{a}_{b c d} y02{b} y02{c} y02{d}
- (1/24) Q{a}_{b c d e} y01{b} y01{c} y01{d} y01{e}
- (1/120) Q{a}_{b c d e f} y01{b} y01{c} y01{d} y01{e} y01{f}:

@substitute!(coord)(y01{a} -> @(y01), y02{a} -> @(y02), y03{a} -> @(y03), y05{a} -> @(y05)):
@distribute!(%): @canonicalise!(%):

# --- now truncate at 5-th order in the curvatures -----
# Q{a}_{...} with n-lower indices contributes  $\backslash\text{eps}^n$  in the curvature expansion.
# We will retain terms to  $\backslash\text{eps}^5$ 

poly:=@(coord):

Q{a}_{b c}::Weight(label=Qterms,value=2).
Q{a}_{b c d}::Weight(label=Qterms,value=3).
Q{a}_{b c d e}::Weight(label=Qterms,value=4).
Q{a}_{b c d e f}::Weight(label=Qterms,value=5).

term00:=@(poly): @keep_weight!(term00){Qterms}{0}:
term01:=@(poly): @keep_weight!(term01){Qterms}{1}:
term02:=@(poly): @keep_weight!(term02){Qterms}{2}:
term03:=@(poly): @keep_weight!(term03){Qterms}{3}:
term04:=@(poly): @keep_weight!(term04){Qterms}{4}:
term05:=@(poly): @keep_weight!(term05){Qterms}{5}:

```

```

coord:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04) + @(term05):
@prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@print["x^a(s)=",~@(coord)~"+\BigO{\eps^6}"];

# =====
#   the x^a(s) in terms of the generalised Gammas
# =====


$$x^a(s) = \left( x^a + Dx^a s + \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c s - \frac{1}{2} Q^a{}_{bc} Dx^b Dx^c ss + \frac{1}{6} Q^a{}_{bcd} Dx^b Dx^c Dx^d s - \frac{1}{6} Q^a{}_{bcd} Dx^b Dx^c Dx^d sss + \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Dx^c Dx^d Dx^e s + \frac{1}{24} Q^a{}_{bcde} Dx^b Dx^c Dx^d Dx^e s \right. \\ \left. - \frac{1}{2} Q^a{}_{bc} Q^b{}_{de} Dx^c Dx^d Dx^e ss - \frac{1}{24} Q^a{}_{bcde} Dx^b Dx^c Dx^d Dx^e ssss + \frac{1}{6} Q^a{}_{bc} Q^b{}_{def} Dx^c Dx^d Dx^e Dx^f s + \frac{1}{4} Q^b{}_{cd} Q^a{}_{bef} Dx^c Dx^d Dx^e Dx^f s \right. \\ \left. + \frac{1}{120} Q^a{}_{bcdef} Dx^b Dx^c Dx^d Dx^e Dx^f s - \frac{1}{6} Q^a{}_{bc} Q^b{}_{def} Dx^c Dx^d Dx^e Dx^f ss - \frac{1}{4} Q^b{}_{cd} Q^a{}_{bef} Dx^c Dx^d Dx^e Dx^f sss - \frac{1}{120} Q^a{}_{bcdef} Dx^b Dx^c Dx^d Dx^e Dx^f sssss \right) + \mathcal{O}(\epsilon^6)$$


# =====
#   now express x^a(s) in terms of the Riemann curvatures
# =====

# the QQ1, QQ2 and QQ3 are *identical* to those previously calculated

@substitute!(coord)(Q^{a}_{b c} Q^{b}_{d e} -> QQ1^{a}_{c d e},
                    Q^{a}_{b c} Q^{b}_{d e f} -> QQ2^{a}_{c d e f},
                    Q^{b}_{c d} Q^{a}_{b e f} -> QQ3^{a}_{c d e f}):

@substitute!(coord)(Q^{a}_{b c} -> @(genGamma02),
                    Q^{a}_{b c d} -> @(genGamma03),
                    Q^{a}_{b c d e} -> @(genGamma04),
                    Q^{a}_{b c d e f} -> @(genGamma05),
                    QQ1^{a}_{c d e} -> @(QQ1),
                    QQ2^{a}_{c d e f} -> @(QQ2),
                    QQ3^{a}_{c d e f} -> @(QQ3)):

@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

# =====
#   in an effort to tidy up the equations we will expand each term0* in powers of s and then DX^a
# =====

```

```

poly:=@(coord):

s::Weight(label=sterms,value=1).

term00:=@(poly): @keep_weight!(term00){sterms}{0}:
term01:=@(poly): @keep_weight!(term01){sterms}{1}:
term02:=@(poly): @keep_weight!(term02){sterms}{2}:
term03:=@(poly): @keep_weight!(term03){sterms}{3}:
term04:=@(poly): @keep_weight!(term04){sterms}{4}:
term05:=@(poly): @keep_weight!(term05){sterms}{5}:

{Dx^{a},x^{a},R^{a}_{b c d}}::SortOrder.

@prodsort!(term01): @rename_dummies!(%): @substitute!(%)(s -> 1):
@prodsort!(term02): @rename_dummies!(%): @substitute!(%)(s -> 1):
@prodsort!(term03): @rename_dummies!(%): @substitute!(%)(s -> 1):
@prodsort!(term04): @rename_dummies!(%): @substitute!(%)(s -> 1):
@prodsort!(term05): @rename_dummies!(%): @substitute!(%)(s -> 1):

# =====
#   now expand each term0* in powers of DX^a, then rebuild term0* and factor out Dx^a
# =====

# expand term01 in powers of Dx^a
poly:=@(term01):

Dx^{a}::Weight(label=Dxterms,value=1).

term0100:=@(poly): @keep_weight!(term0100){Dxterms}{0}:
term0101:=@(poly): @keep_weight!(term0101){Dxterms}{1}:
term0102:=@(poly): @keep_weight!(term0102){Dxterms}{2}:
term0103:=@(poly): @keep_weight!(term0103){Dxterms}{3}:
term0104:=@(poly): @keep_weight!(term0104){Dxterms}{4}:
term0105:=@(poly): @keep_weight!(term0105){Dxterms}{5}:

@factor_out!!(term0100){Dx^{a}}:
@factor_out!!(term0101){Dx^{a}}:
@factor_out!!(term0102){Dx^{a}}:

```

```

@factor_out!!(term0103){Dx^{a}}:
@factor_out!!(term0104){Dx^{a}}:
@factor_out!!(term0105){Dx^{a}}:

term01:=@(term0100) + @(term0101) + @(term0102) + @(term0103) + @(term0104) + @(term0105):
# expand term02 in powers of Dx^a
poly:=@(term02):

Dx^{a}::Weight(label=Dxterms,value=1).

term0200:=@(poly): @keep_weight!(term0200){Dxterms}{0}:
term0201:=@(poly): @keep_weight!(term0201){Dxterms}{1}:
term0202:=@(poly): @keep_weight!(term0202){Dxterms}{2}:
term0203:=@(poly): @keep_weight!(term0203){Dxterms}{3}:
term0204:=@(poly): @keep_weight!(term0204){Dxterms}{4}:
term0205:=@(poly): @keep_weight!(term0205){Dxterms}{5}:

@factor_out!!(term0200){Dx^{a}}:
@factor_out!!(term0201){Dx^{a}}:
@factor_out!!(term0202){Dx^{a}}:
@factor_out!!(term0203){Dx^{a}}:
@factor_out!!(term0204){Dx^{a}}:
@factor_out!!(term0205){Dx^{a}}:

term02:=@(term0200) + @(term0201) + @(term0202) + @(term0203) + @(term0204) + @(term0205):
# expand term03 in powers of Dx^a
poly:=@(term03):

Dx^{a}::Weight(label=Dxterms,value=1).

term0300:=@(poly): @keep_weight!(term0300){Dxterms}{0}:
term0301:=@(poly): @keep_weight!(term0301){Dxterms}{1}:
term0302:=@(poly): @keep_weight!(term0302){Dxterms}{2}:
term0303:=@(poly): @keep_weight!(term0303){Dxterms}{3}:
term0304:=@(poly): @keep_weight!(term0304){Dxterms}{4}:
term0305:=@(poly): @keep_weight!(term0305){Dxterms}{5}:

```

```

@factor_out!!(term0300){Dx^{a}}:
@factor_out!!(term0301){Dx^{a}}:
@factor_out!!(term0302){Dx^{a}}:
@factor_out!!(term0303){Dx^{a}}:
@factor_out!!(term0304){Dx^{a}}:
@factor_out!!(term0305){Dx^{a}}:

term03:=@(term0300) + @(term0301) + @(term0302) + @(term0303) + @(term0304) + @(term0305):
# expand term04 in powers of Dx^a
poly:=@(term04):

Dx^{a}::Weight(label=Dxterms,value=1).

term0400:=@(poly): @keep_weight!(term0400){Dxterms}{0}:
term0401:=@(poly): @keep_weight!(term0401){Dxterms}{1}:
term0402:=@(poly): @keep_weight!(term0402){Dxterms}{2}:
term0403:=@(poly): @keep_weight!(term0403){Dxterms}{3}:
term0404:=@(poly): @keep_weight!(term0404){Dxterms}{4}:
term0405:=@(poly): @keep_weight!(term0405){Dxterms}{5}:

@factor_out!!(term0400){Dx^{a}}:
@factor_out!!(term0401){Dx^{a}}:
@factor_out!!(term0402){Dx^{a}}:
@factor_out!!(term0403){Dx^{a}}:
@factor_out!!(term0404){Dx^{a}}:
@factor_out!!(term0405){Dx^{a}}:

term04:=@(term0400) + @(term0401) + @(term0402) + @(term0403) + @(term0404) + @(term0405):
# expand term05 in powers of Dx^a
poly:=@(term05):

Dx^{a}::Weight(label=Dxterms,value=1).

term0500:=@(poly): @keep_weight!(term0500){Dxterms}{0}:
term0501:=@(poly): @keep_weight!(term0501){Dxterms}{1}:

```

```

term0502:=@(poly): @keep_weight!(term0502){Dxterms}{2}:
term0503:=@(poly): @keep_weight!(term0503){Dxterms}{3}:
term0504:=@(poly): @keep_weight!(term0504){Dxterms}{4}:
term0505:=@(poly): @keep_weight!(term0505){Dxterms}{5}:

@factor_out!!(term0500){Dx^{a}}:
@factor_out!!(term0501){Dx^{a}}:
@factor_out!!(term0502){Dx^{a}}:
@factor_out!!(term0503){Dx^{a}}:
@factor_out!!(term0504){Dx^{a}}:
@factor_out!!(term0505){Dx^{a}}:

term05:=@(term0500) + @(term0501) + @(term0502) + @(term0503) + @(term0504) + @(term0505):

# all done, now print the terms of x^a(s)

@substitute!(term00)(Dx^{a} -> \Delta{x}^{a}); "term00.trn"
@substitute!(term01)(Dx^{a} -> \Delta{x}^{a}); "term01.trn"
@substitute!(term02)(Dx^{a} -> \Delta{x}^{a}); "term02.trn"
@substitute!(term03)(Dx^{a} -> \Delta{x}^{a}); "term03.trn"
@substitute!(term04)(Dx^{a} -> \Delta{x}^{a}); "term04.trn"
@substitute!(term05)(Dx^{a} -> \Delta{x}^{a}); "term05.trn"

@print["\Btag{01}x^a_0="~@(term00)~"\Etag{01}"];
@print["\Btag{02}x^a_1="~@(term01)~"\Etag{02}"];
@print["\Btag{03}x^a_2="~@(term02)~"\Etag{03}"];
@print["\Btag{04}x^a_3="~@(term03)~"\Etag{04}"];
@print["\Btag{05}x^a_4="~@(term04)~"\Etag{05}"];
@print["\Btag{06}x^a_5="~@(term05)~"\Etag{06}"];

```



The solution to the geodesic boundary value problem is

$$x^a(s) = x_0^a + x_1^a s + x_2^a s^2 + x_3^a s^3 + x_4^a s^4 + x_5^a s^5 + \mathcal{O}(\epsilon^6)$$

with

$$x_0^a = x^a$$

$$\begin{aligned} x_1^a = & \left( \Delta x^a + \Delta x^b \Delta x^c \left( \frac{1}{3} x^d R^a{}_{bdc} + \frac{1}{12} x^d x^e \nabla_b R^a{}_{dec} + \frac{1}{6} x^d x^e \nabla_d R^a{}_{bec} + \frac{1}{24} x^d x^e \nabla^a R_{dbec} + \frac{4}{45} x^d x^e x^f R^a{}_{dge} R_{gbfc} - \frac{2}{45} x^d x^e x^f R^a{}_{bgd} R_{gefc} - \frac{1}{45} x^d x^e x^f R^a{}_{dgb} R_{gefc} \right. \right. \\ & + \frac{1}{20} x^d x^e x^f \nabla_{db} R^a{}_{efc} + \frac{1}{20} x^d x^e x^f \nabla_{de} R^a{}_{bfc} + \frac{1}{45} x^d x^e x^f R^a{}_{gdb} R_{gefc} + \frac{1}{40} x^d x^e x^f \nabla_d R_{ebfc} + \frac{2}{45} x^d x^e x^f x^g R_{hbd} \nabla_e R^a{}_{fhg} + \frac{1}{60} x^d x^e x^f x^g R^a{}_{dhe} \nabla_b R_{hfgc} \\ & + \frac{2}{45} x^d x^e x^f x^g R^a{}_{dhe} \nabla_f R_{hbgc} - \frac{1}{45} x^d x^e x^f x^g R^a{}_{bhd} \nabla_e R_{hfgc} - \frac{1}{90} x^d x^e x^f x^g R^a{}_{dhb} \nabla_e R_{hfgc} - \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_c R^a{}_{fhg} - \frac{1}{45} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{chg} \\ & - \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{ghc} + \frac{1}{60} x^d x^e x^f x^g \nabla_{deb} R^a{}_{fgc} + \frac{1}{90} x^d x^e x^f x^g \nabla_{def} R^a{}_{bgc} + \frac{1}{72} x^d x^e x^f x^g R^a{}_{dhe} \nabla_h R_{fbgc} + \frac{1}{90} x^d x^e x^f x^g R^a{}_{hdb} \nabla_e R_{hfgc} \\ & - \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla^a R_{hfgc} + \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{hgc} + \frac{1}{120} x^d x^e x^f x^g \nabla_{de} R_{fbgc} \Big) + \Delta x^b \Delta x^c \Delta x^d \left( \frac{1}{12} x^e \nabla_b R^a{}_{ced} + \frac{1}{45} x^e x^f R^a{}_{egb} R_{gcfd} \right. \\ & + \frac{4}{45} x^e x^f R^a{}_{bge} R_{gcfd} - \frac{1}{45} x^e x^f R^a{}_{bgc} R_{gefd} + \frac{1}{20} x^e x^f \nabla_{eb} R^a{}_{cfd} + \frac{1}{60} x^e x^f \nabla_{bc} R^a{}_{efd} + \frac{7}{45} x^e x^f R^a{}_{geb} R_{gcfd} + \frac{1}{120} x^e x^f \nabla^a{}_b R_{ecfd} + \frac{1}{54} x^e x^f x^g R_{hbec} \nabla_d R^a{}_{fhg} \\ & + \frac{1}{27} x^e x^f x^g R_{hbec} \nabla_h R^a{}_{fgd} + \frac{1}{27} x^e x^f x^g R_{hbec} \nabla_f R^a{}_{dhg} + \frac{2}{27} x^e x^f x^g R_{hbec} \nabla_f R^a{}_{hgd} - \frac{1}{108} x^e x^f x^g R_{hbec} \nabla^a R_{hfgd} + \frac{1}{54} x^e x^f x^g R^a{}_{bhe} \nabla_c R_{hfgd} \\ & + \frac{1}{27} x^e x^f x^g R^a{}_{bhe} \nabla_f R_{hcgd} + \frac{1}{108} x^e x^f x^g R^a{}_{bhe} \nabla_h R_{fcgd} + \frac{1}{27} x^e x^f x^g R^a{}_{heb} \nabla_c R_{hfgd} + \frac{2}{27} x^e x^f x^g R^a{}_{heb} \nabla_f R_{hcgd} + \frac{1}{54} x^e x^f x^g R^a{}_{heb} \nabla_h R_{fcgd} \Big) \\ & + \Delta x^b \Delta x^c \Delta x^d \Delta x^e \left( \frac{1}{54} x^f R^a{}_{bgc} R_{gdfe} + \frac{1}{216} x^f x^g R^a{}_{bhc} \nabla_d R_{hfgc} + \frac{1}{108} x^f x^g R^a{}_{bhc} \nabla_f R_{hdgc} + \frac{1}{432} x^f x^g R^a{}_{bhc} \nabla_h R_{fdgc} + \frac{1}{216} x^f x^g R_{hbf} \nabla_d R^a{}_{ghe} \right. \\ & - \frac{5}{216} x^f x^g R_{hbf} \nabla_d R^a{}_{hgc} + \frac{1}{108} x^f x^g R_{hbf} \nabla_g R^a{}_{dhe} + \frac{7}{108} x^f x^g R_{hbf} \nabla_h R^a{}_{dgc} + \frac{1}{216} x^f x^g R_{hbf} \nabla^a R_{hdgc} + \frac{1}{36} x^f x^g R_{hbf} \nabla_d R^a{}_{ehg} + \frac{1}{72} x^f x^g R^a{}_{bhf} \nabla_c R_{hdgc} \\ & \left. + \frac{1}{24} x^f x^g R^a{}_{hfb} \nabla_c R_{hdgc} \right) + \Delta x^b \Delta x^c \Delta x^d \Delta x^e \Delta x^f \left( \frac{1}{180} x^g R^a{}_{bhc} \nabla_d R_{hegf} + \frac{1}{120} x^g R_{hbgc} \nabla_d R^a{}_{ehf} \right) \Big) \end{aligned}$$

$$\begin{aligned}
x_2^a = & \left( \Delta x^b \Delta x^c \left( -\frac{1}{3} x^d R^a{}_{bdc} - \frac{1}{12} x^d x^e \nabla_b R^a{}_{dec} - \frac{1}{6} x^d x^e \nabla_d R^a{}_{bec} - \frac{1}{24} x^d x^e \nabla^a R_{dbec} - \frac{4}{45} x^d x^e x^f R^a{}_{dge} R_{gbfc} + \frac{2}{45} x^d x^e x^f R^a{}_{bgd} R_{gefc} + \frac{1}{45} x^d x^e x^f R^a{}_{dgb} R_{gefc} \right. \right. \\
& - \frac{1}{20} x^d x^e x^f \nabla_{db} R^a{}_{efc} - \frac{1}{20} x^d x^e x^f \nabla_{de} R^a{}_{bfc} - \frac{1}{45} x^d x^e x^f R^a{}_{gdb} R_{gefc} - \frac{1}{40} x^d x^e x^f \nabla^a R_{ebfc} - \frac{2}{45} x^d x^e x^f x^g R_{hbd} \nabla_e R^a{}_{fhg} - \frac{1}{60} x^d x^e x^f x^g R^a{}_{dhe} \nabla_b R_{hfgc} \\
& - \frac{2}{45} x^d x^e x^f x^g R^a{}_{dhe} \nabla_f R_{hbgc} + \frac{1}{45} x^d x^e x^f x^g R^a{}_{bhd} \nabla_e R_{hfgc} + \frac{1}{90} x^d x^e x^f x^g R^a{}_{dhh} \nabla_e R_{hfgc} + \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_c R^a{}_{fhg} + \frac{1}{45} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{chg} \\
& + \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{ghc} - \frac{1}{60} x^d x^e x^f x^g \nabla_{deb} R^a{}_{fgc} - \frac{1}{90} x^d x^e x^f x^g \nabla_{def} R^a{}_{bgc} - \frac{1}{72} x^d x^e x^f x^g R^a{}_{dhe} \nabla_h R_{fbgc} - \frac{1}{90} x^d x^e x^f x^g R^a{}_{hdb} \nabla_e R_{hfgc} \\
& + \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla^a R_{hfgc} - \frac{1}{90} x^d x^e x^f x^g R_{hdeb} \nabla_f R^a{}_{hgc} - \frac{1}{120} x^d x^e x^f x^g \nabla^a R_{fbgc} \Big) + \Delta x^b \Delta x^c \Delta x^d \left( -\frac{2}{9} x^e x^f R^a{}_{geb} R^g{}_{cfd} - \frac{1}{18} x^e x^f x^g R^a{}_{heb} \nabla_c R^h{}_{fgd} \right. \\
& - \frac{1}{9} x^e x^f x^g R^a{}_{heb} \nabla_f R^h{}_{cgd} - \frac{1}{36} x^e x^f x^g R^a{}_{heb} \nabla^h R_{fcgd} - \frac{1}{18} x^e x^f x^g R^h{}_{bec} \nabla_h R^a{}_{fgd} - \frac{1}{9} x^e x^f x^g R^h{}_{bec} \nabla_f R^a{}_{hgd} + \frac{1}{36} x^e x^f x^g R^h{}_{bec} \nabla^a R_{hfgd} \Big) \\
& \left. - \frac{1}{18} \Delta x^b \Delta x^c \Delta x^d \Delta x^e x^f x^g R^a{}_{hfb} \nabla_c R^h{}_{dge} \right)
\end{aligned}$$

$$\begin{aligned}
x_3^a = & \left( \Delta x^b \Delta x^c \Delta x^d \left( -\frac{1}{12} x^e \nabla_b R^a{}_{ced} - \frac{1}{45} x^e x^f R^a{}_{egb} R_{gcfd} - \frac{4}{45} x^e x^f R^a{}_{bge} R_{gcfd} + \frac{1}{45} x^e x^f R^a{}_{bgc} R_{gefd} - \frac{1}{20} x^e x^f \nabla_{eb} R^a{}_{cfd} - \frac{1}{60} x^e x^f \nabla_{bc} R^a{}_{efd} \right. \right. \\
& + \frac{1}{15} x^e x^f R^a{}_{geb} R_{gcfd} - \frac{1}{120} x^e x^f \nabla^a R_{ecfd} - \frac{1}{54} x^e x^f x^g R_{hbec} \nabla_d R^a{}_{fhg} + \frac{1}{54} x^e x^f x^g R_{hbec} \nabla_h R^a{}_{fgd} - \frac{1}{27} x^e x^f x^g R_{hbec} \nabla_f R^a{}_{dhg} + \frac{1}{27} x^e x^f x^g R_{hbec} \nabla_f R^a{}_{hgd} \\
& - \frac{1}{54} x^e x^f x^g R_{hbec} \nabla^a R_{hfgd} - \frac{1}{54} x^e x^f x^g R^a{}_{bhe} \nabla_c R_{hfgd} - \frac{1}{27} x^e x^f x^g R^a{}_{bhe} \nabla_f R_{hcgd} - \frac{1}{108} x^e x^f x^g R^a{}_{bhe} \nabla_h R_{fcgd} + \frac{1}{54} x^e x^f x^g R^a{}_{heb} \nabla_c R_{hfgd} \\
& \left. + \frac{1}{27} x^e x^f x^g R^a{}_{heb} \nabla_f R_{hcgd} + \frac{1}{108} x^e x^f x^g R^a{}_{heb} \nabla_h R_{fcgd} \right) - \frac{1}{12} \Delta x^b \Delta x^c \Delta x^d \Delta x^e x^f x^g R^h{}_{bfc} \nabla_h R^a{}_{dge} \Big)
\end{aligned}$$

$$\begin{aligned}
x_4^a = & \Delta x^b \Delta x^c \Delta x^d \Delta x^e \left( -\frac{1}{54} x^f R^a{}_{bgc} R_{gdfe} - \frac{1}{216} x^f x^g R^a{}_{bhc} \nabla_d R_{hfgc} - \frac{1}{108} x^f x^g R^a{}_{bhc} \nabla_f R_{hdgc} - \frac{1}{432} x^f x^g R^a{}_{bhc} \nabla_h R_{fdgc} - \frac{1}{216} x^f x^g R_{hbfc} \nabla_d R^a{}_{ghe} \right. \\
& + \frac{5}{216} x^f x^g R_{hbfc} \nabla_d R^a{}_{hgc} - \frac{1}{108} x^f x^g R_{hbfc} \nabla_g R^a{}_{dhe} + \frac{1}{54} x^f x^g R_{hbfc} \nabla_h R^a{}_{dgc} - \frac{1}{216} x^f x^g R_{hbfc} \nabla^a R_{hdgc} - \frac{1}{36} x^f x^g R_{hbfc} \nabla_d R^a{}_{ehg} - \frac{1}{72} x^f x^g R^a{}_{bhf} \nabla_c R_{hdgc} \\
& \left. + \frac{1}{72} x^f x^g R^a{}_{hfb} \nabla_c R_{hdgc} \right)
\end{aligned}$$

$$x_5^a = \Delta x^b \Delta x^c \Delta x^d \Delta x^e \Delta x^f \left( -\frac{1}{180} x^g R^a{}_{bhc} \nabla_d R_{hegf} - \frac{1}{120} x^g R_{hbgc} \nabla_d R^a{}_{ehf} \right)$$

```

# here is a check : evaluate x^{a}(s) at s=1, this should give x^{a} + Dx^{a}, yes, it does!
tmp:=@(term00) + @(term01) + @(term02) + @(term03) + @(term04) + @(term05):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%);

```

$$tmp := x^a + \Delta x^a$$

```

# =====
#   export
# =====

com:="open geodesic-bvp.lib":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y01Gamma.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y02Gamma.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y03Gamma.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y04Gamma.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y05Gamma.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib y05Riemann.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib QQ1.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib QQ2.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib QQ3.txt":
@run(com){"/Users/leo/local/sh/cdbfile"}:

# =====
#   export the solution of the bvp for use by reformat*.cdbp
# =====

com:="export geodesic-bvp.lib term00.trn":

```

```
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib term01.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib term02.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib term03.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib term04.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-bvp.lib term05.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:
```